# PME 3380 – Modelagem de Sistemas Dinâmicos

## LISTA D

Tiago Vieira de Campos Krause 9836238

Exercícios:

1 - Faça as modificações adequadas para se poder desenhar e comparar os gráficos da resposta do sistema não linear e linear. Faça as simulações dos sistemas linear e não linear considerando que o reservatório parte do nível $h = 2$ m, mas com vazão de entrada nula. Compare as respostas.

Utilizando o seguinte código adaptado para plotar os resultados das simulações dos sistemas linear e não linear e comparando o nível do reservatório obtido.

```
// Simulacao de sistema linear e não linear
// É sempre melhor apagar as variaveis anteriores
clear all
clc

// Carregar a funcao que implementa o modelo matematico do sistema
exec("D:\Poli\8 semestre\Modelagem\Lista 4\reservatorio.sci");
// Definir parametros:
S=10; // [m^2] Area da secao transversal do reservatorio
rho=1000; // [kg/m^3] massa especifica da agua
g=10; // [m/s^2] aceleração da gravidade na superficie da Terra
R=2*10^8; // [Pa/(m^3/s)^2] parametro que relaciona pressao e vazao
ho=2; // [m] nivel do reservatorio em regime
hi=-2; // [m] nivel adicional desejado
Qei_linearizada=(1/2)*sqrt(rho*g/(R*ho))*hi; // [m^3/s] vazao na entrada
Qei=sqrt(rho*g*(ho+hi)/R); // [m^3/s] vazao na entrada

// Definir a condicao inicial:
h0=2; // [m] nivel do reservatorio na condicao inicial
x0=0; // [m] desvio inicial do nivel em relação ao equilibrio

// Definir o vetor de instantes de tempo:
t=0:10:40000;
// Definir o vetor de entradas:
u=Qei_linearizada*ones(t);

// Definir o sistema não linear usando a função ode:
h=ode(h0,t(1),t,list(tanque,entrada)); // h é o nivel do reservatorio [m]

// Definir o sistema linear usando o comando syslin:
A=(-1/(2*S))*sqrt(rho*g/(R*ho));
B=1/S;
C=1;
D=0;
tanque_linearizado=syslin('c',A,B,C,D); // o parametro 'c' indica que o sistema
                                        // é continuo no tempo
```

```
// Simulando o sistema usando o comando csim:
[y,x]=csim(u,t,tanque_linearizado,x0);
// Plotando o resultado em verde:
plot2d(t,y+h0,3)
plot2d(t,h,2)
// Colocando um titulo na figura e nomeando os eixos:
xtitle("Resposta do tanque","Tempo t [s]","Nível do reservatório [m]");
// Colocando legendas:
legends(["Nível do reservatório usando sistema linear",
"Nível do reservatório usando sistema não linear"], [3,2], "below")
// Colocando uma grade azul no grafico:
xgrid(2)
```

```
// Definicao da funcao que implementa a equacao nao linear
function [hdot]=tanque(t,h,Qe)
hdot=(-sqrt(rho*g*h/R)+Qe(t))/S
endfunction

// Definicao da funcao que implementa a entrada Qe:
function [u]=entrada(t)
u=Qei;
// supondo o exemplo, u=K1*sin(w*t)+K2*t^(-2)
//u=0.0005*sin(0.001*t)+0.0085
endfunction
```

*Códigos utilizados*




*Gráfico plotado*

A diferença entre os resultados quando a vazão de entrada é nula se torna mais significativa do que para os casos com vazão positiva. Além disso, variando a vazão de entrada pode-se observar que o equilíbrio é atingido mais rapidamente pelo sistema não linear, que não segue a curva padrão exponencial do sistema linear.

2 - Obtenha o modelo matemático do circuito elétrico mostrado abaixo e compare com o modelo linear do sistema com um reservatório. Faça simulações e compare qualitativamente com os resultados do exercício 1 (sistema linear).




2) Para o circuito a seguir análogo ao sistema do reservatório




Pela Lei de Kirchhof chega-se às seguintes equações:

$$\begin{cases} e_o = e_i - R \cdot i \\ e_o = \frac{q}{C} \end{cases}$$

Como $i = \dot{q}$, então

$$\dot{q} = \frac{e_i}{R} - \frac{q}{RC}$$

Esta equação se assemelha à equação linear do sistema do reservatório obtida anteriormente:

$$\dot{x} = \frac{1}{S}u - \frac{1}{2S}\sqrt{\frac{\rho g}{R h_0}}\, x$$

Sendo que ambos seguem o modelo genérico de EDO linear $\dot{x} = Ax + B$

Através da comparação entre as equações citada acima, define-se os valores de R e C na equação obtida pela analogia elétrica. Pode-se observar que, por seguirem o mesmo modelo $\dot{x} = Ax + B$, define-se R, C, $q$ e $e_i$do sitema elétrico em função dos parâmetros do sistema hidráulico. Deste modo, descreve-se o espaço de estados para a equação para então rodar a simulação através do seguinte código:

```
// Simulacao de sistema linear
// É sempre melhor apagar as variaveis anteriores
clear
clc

// Parametros para o sistema hidráulico:
S=10; // [m^2] Area da secao transversal do reservatorio
rho=1000; // [kg/m^3] massa especifica da agua
g=10; // [m/s^2] aceleração da gravidade na superfície da Terra
R=2*10^8; // [Pa/(m^3/s)^2] parametro que relaciona pressao e vazao
ho=2; // [m] nivel do reservatorio em regime
hi=-2; // [m] nivel adicional desejado
Qei=(1/2)*sqrt(rho*g/(R*ho))*hi; // [m^3/s] vazao na entrada

// Definir a condicao inicial:
q0=0; // [m] desvio inicial do nivel em relação ao equilibrio
// Definir o vetor de instantes de tempo:
t=0:10:40000;
// Definir o vetor de entradas:
u=Qei*ones(t);

// Conversão dos parâmetros para o sistema elétrico análogo:
Ran=S
Can=2*sqrt((R*ho)/(rho*g));
ei=u

// Definir o sistema linear usando o comando syslin:
A=-1/Ran/Can
B=1/Ran;
C=1;
D=0;
tanque=syslin('c',A,B,C,D); // o parametro 'c' indica que o sistema é
 // continuo no tempo

// Simulando o sistema usando o comando csim:
[y,q]=csim(u,t,tanque,q0);
// Plotando o resultado em verde:
plot2d(t,q+ho,3)
// Colocando um titulo na figura e nomeando os eixos:
xtitle("Resposta no sistema elétrico","Tempo t [s]","Carga [C]");
// Colocando uma grade azul no grafico:
xgrid(2)
```

*Código usado para simulação do sistema elétrico*




*Gráfico obtido através do código*

O resultado obtido é igual ao calculado no item anterior uma vez que o espaço de estados descrito foi igual e os métodos de simulação utilizados não se alteraram.

## Lição de casa:

**1 - Usando a abordagem vista nestes exemplos, faça a simulação do sistema com dois reservatórios, supondo o modelo linear:**



Através da equação linearizada do sistema com dois reservatórios calculada na lista anterior e aplicando-a ao seguinte código obtém-se os gráficos a seguir:

$$\left.\frac{\partial \dot{h}_1}{\partial h_1}\right|_{h_{10},h_{20}} = -\frac{1}{2S_1}\sqrt{\frac{\rho g}{R_1(h_{10}-h_{20})}} \qquad \left.\frac{\partial \dot{h}_2}{\partial h_1}\right|_{h_{10},h_{20}} = \frac{1}{2S_2}\sqrt{\frac{\rho g}{R_1(h_{10}-h_{20})}}$$

$$\left.\frac{\partial \dot{h}_1}{\partial h_2}\right|_{h_{10},h_{20}} = \frac{1}{2S_1}\sqrt{\frac{\rho g}{R_1(h_{10}-h_{20})}} \qquad \left.\frac{\partial \dot{h}_2}{\partial h_2}\right|_{h_{10},h_{20}} = -\frac{1}{2S_2}\sqrt{\frac{\rho g}{R_1(h_{10}-h_{20})}} - \frac{1}{2S_2}\sqrt{\frac{\rho g}{R_2 h_{20}}}$$

$$\left.\frac{\partial \dot{h}_1}{\partial Q_e}\right|_{Q_{e0}} = \frac{1}{S_1} \qquad \left.\frac{\partial \dot{h}_2}{\partial Q_e}\right|_{Q_{e0}} = 0$$

$$\underbrace{\begin{bmatrix} \dot{h}_1 \\ \dot{h}_2 \end{bmatrix}}_{\dot{x}} = \underbrace{\begin{bmatrix} \left.\frac{\partial \dot{h}_1}{\partial h_1}\right|_{h_{10},h_{20}} & \left.\frac{\partial \dot{h}_1}{\partial h_2}\right|_{h_{10},h_{20}} \\ \left.\frac{\partial \dot{h}_2}{\partial h_1}\right|_{h_{10},h_{20}} & \left.\frac{\partial \dot{h}_2}{\partial h_2}\right|_{h_{10},h_{20}} \end{bmatrix}}_{A} \underbrace{\begin{bmatrix} h_1 - h_{10} \\ h_2 - h_{20} \end{bmatrix}}_{x} + \underbrace{\begin{bmatrix} \left.\frac{\partial \dot{h}_1}{\partial Q_e}\right|_{Q_{e0}} \\ \left.\frac{\partial \dot{h}_2}{\partial Q_e}\right|_{Q_{e0}} \end{bmatrix}}_{B} \underbrace{(Q_e - Q_{e0})}_{u}$$

$$\underbrace{\begin{bmatrix} y_1 \\ y_2 \end{bmatrix}}_{y} = \underbrace{\begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}}_{C} \underbrace{\begin{bmatrix} x_1 \\ x_2 \end{bmatrix}}_{x} + \underbrace{\begin{bmatrix} 0 \\ 0 \end{bmatrix}}_{D} \underbrace{(Q_e - Q_{e0})}_{u}$$

*Equações do sistema com dois reservatórios linearizado*

```
// Simulacao de sistema linear
// É sempre melhor apagar as variaveis anteriores
clear
clc
// Definir parametros:
S1 = 10 //m2 - área da seção transversal do reservatório 1
S2 = 10 //m2 - área da seção transversal do reservatório 2
R1 = 2*10^8 //Pa/(m3/s)2 - parâmetro que relaciona vazão com queda
             // de pressão (perda de carga) na saída do reservatório 1
R2 = 2*10^8 //Pa/(m3/s)2 - parâmetro que relaciona vazão com queda
             //de pressão (perda de carga) na saída do reservatório 2
rho=1000; // [kg/m^3] massa especifica da agua
g=10; // [m/s^2] aceleração da gravidade na superficie da Terra
ho=[2;1]; // [m] nivel do reservatorio em regime

Qe0=sqrt(rho*g*(ho(1)-ho(2))/R1) // [m3/s] vazão no equilíbrio

Qei=0.01 // [m3/s] vazão de entrada

// Definir o sistema linear usando o comando syslin:
A=[(-1/(2*S1))*sqrt(rho*g/(R1*(ho(1)-ho(2)))),
 (1/(2*S2))*sqrt(rho*g/(R1*(ho(1)-ho(2))));
 (1/(2*S1))*sqrt(rho*g/(R1*(ho(1)-ho(2)))),
 (-1/(2*S2))*sqrt(rho*g/(R1*(ho(1)-ho(2))))-(1/(2*S2))*sqrt(rho*g/(R2*ho(2)))]
B=[1/S1;0]
C=[1, 0; 0, 1];
D=[0; 0];
tanque=syslin('c',A,B,C,D); // o parametro 'c' indica que o sistema eh
 // continuo no tempo
// Definir a condicao inicial:
x0=[1;0]; // [m] desvio inicial do nivel em relação ao equilibrio
// Definir o vetor de instantes de tempo:
t=0:10:40000;
// Definir o vetor de entradas:
u=Qei*ones(t);
```

```
// Simulando o sistema usando o comando csim:
[y,x]=csim(u,t,tanque,x0);
// Plotando o resultado em verde para o reservatório 1 e azul para o reservatório 2:
plot2d(t,y(1,:),3)
plot2d(t,y(2,:),2)
// Colocando um titulo na figura e nomeando os eixos:
xtitle("Resposta do tanque","Tempo t [s]","Variacao de nivel [m]");
// Colocando legendas:
legends(["Nível do resrevatório 1", "Nível do reservatório 2"],[3,2],"below");
// Colocando uma grade azul no grafico:
xgrid(2)
```

*Código usado para a simulação do sistema*




*Gráfico obtido pelo código acima*

**2 - Desenvolva um circuito elétrico análogo ao sistema com dois reservatórios.**

2) Circuito elétrico análogo ao sistema hidráulico de dois reservatórios: